FLPPF

PURPOSE

Compute the standard form of the fatigue-life (also known as the Birnbaum-Saunders) percent point function.

DESCRIPTION

The fatigue-life distribution can be defined as the average of the inverse gaussian distribution and the reciprocal inverse Gaussian distribution (that is, (1/2)*(igpdf(x,γ) + rigpdf(x,γ)) where igpdf and rigpdf are the probability density functions for these distributions and γ is the shape parameter. This reduces to:

$$f(x) = \left(\frac{1+x}{2}\right) igpdf(x,\gamma) \qquad \text{for } x > 0 \qquad \textbf{(EQ 8-216)}$$

There is no simple closed form for the percent point function. It is calculated numerically using a bisection method.

SYNTAX

LET <y2> = FLPPF(<y1>,<gamma>) <SUBSET/EXCEPT/FOR qualification>

where <y1> is a variable, a number, or a parameter in the range 0 to 1;

<y2> is a variable or a parameter (depending on what <y1> is) where the computed fatigue-life ppf values is stored;

<gamma> is a positive integer (the shape parameter);

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

EXAMPLES

LET A = FLPPF(0.9,10)

LET Y = FLPPF(P,10)

NOTE 1

The fatigue-life distribution is nearly symmetric and moderate tailed for small gamma. It is highly skewed and long tailed for large gamma. It approaches normality as gamma approaches zero.

NOTE 2

The general form of the fatigue-life distribution has a location parameter μ and a scale parameter β. The location parameter defaults to 1 (unlike most distributions where it defaults to zero) and must be positive. See topic (3) under the General considerations section at the beginning of this chapter for a discussion of generating percent point function values for the general form of the distribution.

DEFAULT

None

SYNONYMS

None

RELATED COMMANDS

FLPDF	=	Compute the fatigue-life probability density function.
FLCDF	=	Compute the fatigue-life cumulative distribution function.
IGPDF	=	Compute the inverse Gaussian probability density function.
IGPPF	=	Compute the inverse Gaussian percent point function.
IGCDF	=	Compute the inverse Gaussian cumulative distribution function.
RIGPDF	=	Compute the reciprocal inverse Gaussian probability density function.
RIGPPF	=	Compute the reciprocal inverse Gaussian percent point function.
RIGCDF	=	Compute the reciprocal inverse Gaussian cumulative distribution function.

REFERENCE

"Continuous Univariate Distributions - 1," Johnson and Kotz, Houghton Mifflin, 1970 (chapter 15).

"Methods for Statistical Analysis of Reliability and Life Data," Mann, Schaffer, and Singpurwalla, Wiley, 1974 (pp. 150-155).

APPLICATIONS

Reliability

IMPLEMENTATION DATE

90/5

PROGRAM

```
SEGMENT 1 COORDINATES 16 88 21 88; SEGMENT 1 PATTERN SOLID
SEGMENT 2 COORDINATES 16 84 21 84; SEGMENT 2 PATTERN DASH
SEGMENT 3 COORDINATES 16 80 21 80; SEGMENT 3 PATTERN DOT
SEGMENT 4 COORDINATES 16 76 21 76; SEGMENT 4 PATTERN DA2
LEGEND 1 GAMMA = 1; LEGEND 1 COORDINATES 22 87
LEGEND 2 GAMMA = 2; LEGEND 2 COORDINATES 22 83
LEGEND 3 GAMMA = 5; LEGEND 3 COORDINATES 22 79
LEGEND 4 GAMMA = 0.5; LEGEND 4 COORDINATES 22 75
XLIMITS 0 1
XTIC DECIMAL 1
MAJOR XTIC NUMBER 6
MINOR XTIC NUMBER 1
TITLE PLOT FLPPF FOR VARIOUS VALUES OF GAMMA
X1LABEL PROBABILITY; Y1LABEL X
LINES SOLID DASH DOT DASH2
YLIMITS 0 5
MAJOR YTIC MARK NUMBER 6
YTIC OFFSET 0 0.3
PLOT FLPPF(X,1) FOR X = 0.01 .01 0.99 AND
PLOT FLPPF(X,2) FOR X = 0.01 .01 0.90 AND
PLOT FLPPF(X,5) FOR X = 0.01 .01 0.70 AND
PLOT FLPPF(X,0.5) FOR X = 0.01 .01 0.90
```